NAG Library Routine Document

G01ERF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01ERF returns the probability associated with the lower tail of the von Mises distribution between $-\pi$ and π through the function name.

2 Specification

```
FUNCTION G01ERF (T, VK, IFAIL)
REAL (KIND=nag_wp) G01ERF
INTEGER            IFAIL
REAL (KIND=nag_wp) T, VK
```

3 Description

The von Mises distribution is a symmetric distribution used in the analysis of circular data. The lower tail area of this distribution on the circle with mean direction $\mu_0 = 0$ and concentration parameter kappa, κ, can be written as

$$\Pr(\Theta \leq \theta : \kappa) = \frac{1}{2\pi I_0(\kappa)} \int_{-\pi}^{\theta} e^{\kappa \cos \Theta} d\Theta,$$

where θ is reduced modulo 2π so that $-\pi \leq \theta < \pi$ and $\kappa \geq 0$. Note that if $\theta = \pi$ then G01ERF returns a probability of 1. For very small κ the distribution is almost the uniform distribution, whereas for $\kappa \to \infty$ all the probability is concentrated at one point.

The method of calculation for small κ involves backwards recursion through a series expansion in terms of modified Bessel functions, while for large κ an asymptotic Normal approximation is used.

In the case of small κ the series expansion of $\Pr(\Theta \leq \theta \colon \kappa)$ can be expressed as

$$\Pr(\Theta \leq \theta : \kappa) = \tfrac{1}{2} + \frac{\theta}{(2\pi)} + \frac{1}{\pi I_0(\kappa)} \sum_{n=1}^{\infty} n^{-1} I_n(\kappa) \sin n\theta,$$

where $I_n(\kappa)$ is the modified Bessel function. This series expansion can be represented as a nested expression of terms involving the modified Bessel function ratio R_n,

$$R_n(\kappa) = \frac{I_n(\kappa)}{I_{n-1}(\kappa)}, \qquad n = 1, 2, 3, \ldots,$$

which is calculated using backwards recursion.

For large values of κ (see Section 7) an asymptotic Normal approximation is used. The angle Θ is transformed to the nearly Normally distributed variate Z,

$$Z = b(\kappa) \sin \frac{\Theta}{2},$$

where

$$b(\kappa) = \frac{\sqrt{\frac{2}{\pi}} e^{\kappa}}{I_0(\kappa)}$$

and $b(\kappa)$ is computed from a continued fraction approximation. An approximation to order κ^{-4} of the asymptotic normalizing series for z is then used. Finally the Normal probability integral is evaluated.

For a more detailed analysis of the methods used see Hill (1977).

4 References

Hill G W (1977) Algorithm 518: Incomplete Bessel function I_0: The Von Mises distribution *ACM Trans. Math. Software* **3** 279–284

Mardia K V (1972) *Statistics of Directional Data* Academic Press

5 Parameters

1: T – REAL (KIND=nag_wp) *Input*

On entry: θ, the observed von Mises statistic measured in radians.

2: VK – REAL (KIND=nag_wp) *Input*

On entry: the concentration parameter κ, of the von Mises distribution.

Constraint: $\text{VK} \geq 0.0$.

3: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $\text{VK} < 0.0$ and G01ERF returns 0.

7 Accuracy

G01ERF uses one of two sets of constants depending on the value of ***machine precision***. One set gives an accuracy of six digits and uses the Normal approximation when $\text{VK} \geq 6.5$, the other gives an accuracy of 12 digits and uses the Normal approximation when $\text{VK} \geq 50.0$.

8 Further Comments

Using the series expansion for small κ the time taken by G01ERF increases linearly with κ; for larger κ, for which the asymptotic Normal approximation is used, the time taken is much less.

If angles outside the region $-\pi \leq \theta < \pi$ are used care has to be taken in evaluating the probability of being in a region $\theta_1 \leq \theta \leq \theta_2$ if the region contains an odd multiple of π, $(2n+1)\pi$. The value of $F(\theta_2;\kappa) - F(\theta_1;\kappa)$ will be negative and the correct probability should then be obtained by adding one to the value.

9 Example

This example inputs four values from the von Mises distribution along with the values of the parameter κ. The probabilities are computed and printed.

9.1 Program Text

```
    Program g01erfe

!     G01ERF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g01erf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: p, t, vk
      Integer                          :: ifail
!     .. Executable Statements ..
      Write (nout,*) 'G01ERF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) '      T          VK     Probability'
      Write (nout,*)

d_lp: Do
        Read (nin,*,Iostat=ifail) t, vk
        If (ifail/=0) Then
          Exit d_lp
        End If

!       Calculate probability
        ifail = 0
        p = g01erf(t,vk,ifail)

!       Display the results
        Write (nout,99999) t, vk, p
      End Do d_lp

99999 Format (F10.4,2X,F10.4,2X,F10.4)
    End Program g01erfe
```

9.2 Program Data

```
G01ERF Example Program Data
 7.0  0.0
 2.8  2.4
 1.0  1.0
-1.4  1.3
```

9.3 Program Results

```
 G01ERF Example Program Results

       T          VK     Probability

    7.0000      0.0000      0.6141
    2.8000      2.4000      0.9983
    1.0000      1.0000      0.7944
   -1.4000      1.3000      0.1016
```